C1M10

Using the Derivative to Sketch the Function

The behavior of the derivative reveals a lot about the shape of a curve. Everything we know on this topic depends on one theorem which will be discussed later. Because of the importance of this theorem I would like to touch on it now and at least make you aware of what it says geometrically. Some mathematicians refer to it as *The Fundamental Theorem of Differential Calculus* because so much of what we do in beginning calculus depends on it. This theorem is usually known as *The Mean Value Theorem*. We will use a different format than usual by identifying two hypotheses, H1 and H2, and the conclusion C.

The Mean Value Theorem (MVT):

H1: f is a continuous function on the closed interval $[a, b]$.

H2: f is differentiable on the open interval (a, b).

C: There is a point c in (a, b) that satisfies

$$f\ (c) = \frac{f(b) - f(a)}{b - a}$$

Do you remember how we commented earlier that theorems that assert the existence of some mathematical entity are often very strong theorems? First, let's recognize that the number $\frac{f(b) - f(a)}{b - a}$ is the slope of the line that joins $(a, f(a))$ and $(b, f(b))$. So, the MVT asserts that there is a tangent line to the curve at some interior point that is parallel to this line. Our picture looks like this:



There are a couple of comments that might help here. There may be several points, $c_1, c_2, \ldots c_n$, that satisfy the condition. The point c MUST NOT be an endpoint.

Maple Example: Use Maple to plot the graph of $f(x) = x\sin(3x) + \frac{1}{2}$ on the interval $[1, 3]$ and to draw the line tangent to this curve at the point $(c, f(c))$ where c satisfies the conclusion for the MVT.

Actually, the graph is shown in the diagram above. We will provide the Maple for all but the labeling.

```
> restart:    with(plots):    with(plottools):
> f:=x->x*sin(3*x)+1/2;
```

$$f := x \to x\sin(3x) + \frac{1}{2}$$

```
> a:=1;  b:=3;
```

$$a := 1$$
$$b := 3$$

```
> C:=(f(b)-f(a))/(b-a);
```

$$C := \frac{3}{2}\sin(9) - \frac{1}{2}\sin(3)$$

```
> fprime:=unapply(diff(f(x),x),x);
```

$$fprime := x \to \sin(3x) + 3x\,\cos(3x)$$

```
> c:=fsolve(fprime(x)=C,x,1..3);
```

$$c := 1.672139733$$

```
> m:=fprime(c);  y0:=f(c);
```

$$m := .5476177153$$
$$y0 := -1.095451597$$

```
> eq1:=y-y0=m*(x-c);
```

$$eq1 := y + 1.095451597 = .5476177153\,x - .9156933402$$

```
> y:=solve(eq1,y);
```

$$y := -2.011144937 + .5476177153\,x$$

```
> P1:=plot(f(x),x=a..b,color=red,scaling=constrained):
> P2:=plot(y,x=a-1..b+1,color=blue):
> P3:=line([a,f(a)],[b,f(b)],color=magenta):
> P4:=line([c,0],[c,f(c)],color=green):
> P5:=plot(f(x),x=-.3..(-.1),color=white):
> P6:=line([a,0],[a,f(a)],color=cyan):
> P7:=line([b,0],[b,f(b)],color=cyan):
> display(P1,P2,P3,P4,P5,P6,P7);
```

Suppose that $f\ (x) > 0$ on (a, b). Then $\frac{f(b)-f(a)}{b-a} > 0$ and $f(b) - f(a) > 0$, or $f(a) < f(b)$. But, using the same argument for $a \le x_1 < x_2 \le b$ and applying the MVT on $[x_1, x_2]$ we see that $f(x_1) < f(x_2)$, or in other words, f is *strictly increasing* on $[a, b]$. Similarly, if $f\ (x) < 0$ on $[a, b]$, then f is *strictly decreasing* on $[a, b]$. When the derivative is positive, the function is moving uphill as you move from left to right.

Place your hand so that your fingers are pointing down at about a 45° angle in front of you. Assume that there is an imaginary bar a few inches in front of the tip of your fingers. Now move your hand so that it moves under that bar and upwards in a smooth motion. Imagine the slope of your hand to be -1 as you begin, then the slope increases to 0 as you level out, and then it increases to $+1$ as you stop. It is fair to say that the *derivative function* is *increasing.* If we now let the derivative, f , play the role of f in the MVT, then the derivative of the derivative of f must be strictly positive. Let's take a very simple example as our model. Suppose that $f(x) = x^2$. Then $f\ (x) = 2x$ and $f\ \ (x) = 2$. The derivative is negative when $x < 0$, 0 at $x = 0$ and positive when $x > 0$. Note that $f\ \ (x) > 0$ for all x. A function that behaves like x^2 is called *concave upwards.*

Now for the opposite situation. If our model is like $g(x) = -x^2$, begin with your hand in front of you pointing upwards at 45° and then move your hand smoothly over a bar and downward. The slope was initially about $+1$, it became 0 at the top, and ended at -1. Here the derivative was a *decreasing function*, so the derivative of the derivative must be negative. We say that a function that behaves like $-x^2$ is *concave downwards.*

We will learn later that it is important to discover where a function stops being concave one way and starts being concave the other. Such points occur where the second derivative is zero, or, *does not exist* and are called *inflection points.* Suppose that $p(x) = ax^2 + bx + c$, so that $p\ (x) = 2ax + b$ and $p\ \ (x) = 2a$. Then there is no way this quadratic can have an inflection point - the second derivative is a constant, so that p is always concave up ($a > 0$) or is always concave down ($a < 0$). In order for a polynomial to have an inflection point the degree must be at least three, so only cubics, quartics, quintics,.., and so forth will have points where the concavity changes. In the figure which follows you see a graph of $y = \sqrt[3]{x} = x^{1/3}$. The derivative does not exist at $x = 0$ (the tangent line would be vertical, which is a "no, no"), but the concavity changes from *up* to *down* as we move from left to right.



You probably assume that the graph above was produced by

```
> plot(x^(1/3),x=-3.5..3.5);
```

WRONG!! Maple doesn't like taking fractional powers of negative numbers, and rightly so. In order to get this graph we used `rotate`, which is in `plottools`. First we plotted $y = -x^3$, and then we rotated it 90°.

```
> R:=plot(-x^3,x=-1.5..(1.5),scaling=constrained):
> display(rotate(R,Pi/2));
```

Now let's look at x^2 and $2 - x^2$, and their first and second derivatives.



Concave Upwards



Concave Downwards

Now we will do another Maple example where we illustrate what happens when the second derivative is positive, zero, and negative. We have drawn vertical dotted lines at the points where the second derivative is zero. Please note how the concavity of F changes at these points. Although we did not make special mention of the behavior of F , its slope is zero at these points, as it should be. And the zeroes of F correspond to where our function F crests or bottoms out.

Maple Example: For $F(x) = x^4 - 6x^2 + 3$ on $[-3, 3]$, plot F, its derivative, and its second derivative on the same coordinate system and note the relationships.

```
> restart:    with(plots):    with(plottools):
> F:=x->x^4-6*x^2+3;
```

$$F := x \to x^4 - 6x^2 + 3$$

```
> Fprime:=unapply(diff(F(x),x),x);
```

$$Fprime := x \to 4x^3 - 12x$$

```
> solve(Fprime(x)=0,x);
```

$$0, \sqrt{3}, -\sqrt{3}$$

```
> F2prime:=unapply(diff(Fprime(x),x),x);
```

$$F2prime := x \to 12x^2 - 12$$

```
> solve(F2prime(x)=0,x);
```

$$1, -1$$

```
> S1:=plot(F(x),x=-3..3,color=red):
> S2:=plot(Fprime(x),x=-3..3,color=blue):
> S3:=plot(F2prime(x),x=-3..3,color=green):
> S4:=textplot([-2.5,20.2,"F"]):
> S5:=textplot([-2.5,-48,"F'"]):
> S6:=textplot([-2.2,80,"F''>0"]):
> S61:=textplot([.45,-40,"F''<0"]):
> S62:=textplot([2.2,80,"F''>0"]):
> S7:=line([-1,-70],[-1,90],color=khaki,linestyle=3):
```

```
> S8:=line([1,-70],[1,90],color=khaki,linestyle=3):
> display(S1,S2,S3,S4,S5,S6,S61,S62,S7,S8);
```



C1M10 Problems:

1. Given $f(x) = x^3 - x + 2$ on $[-1, 2]$. Using the first Maple Example as a prototype, plot the graph of $f(x)$ and the tangent line at the point which satisfies the conclusion of the Mean Value Theorem.

2. Given $g(x) = 3x^4 - 4x^3$ on $[-5/4, 2]$. Use Maple to plot $g(x)$, $g\ (x)$, and $g\ \ (x)$.

3. Given $h(x) = \cos(x) + \sin(2x)$ on $[-2, 2]$. Use Maple to plot $h(x)$, $h\ (x)$, and $h\ \ (x)$. Use `fsolve` to find those values where $h\ \ (x)$ is 0 and put vertical lines there. Remember, you can specify the range for the solution in `fsolve`, as we did in one of the examples.